



DECUS

PROGRAM LIBRARY

DECUS NO.	FOCAL8-284
TITLE	8/E EAE ROUTINE FOR FOCAL
AUTHOR	James Van Zee
COMPANY	University of Washington Seattle, Washington
DATE	June 1973
SOURCE LANGUAGE	PAL8

ATTENTION

This is a USER program. Other than requiring that it conform to submittal and review standards, no quality control has been imposed upon this program by DECUS.

The DECUS Program Library is a clearing house only; it does not generate or test programs. No warranty, express or implied, is made by the contributor, Digital Equipment Computer Users Society or Digital Equipment Corporation as to the accuracy or functioning of the program or related material, and no responsibility is assumed by these parties in connection therewith.

8/E EAE ROUTINE FOR FOCAL

DECUS Program Library Write-up

DECUS NO. FOCAL8-284

THIS FOCAL EAE PATCH WAS DESIGNED SPECIFICALLY FOR THE 8/E COMPUTER SERIES. IT USES MODE B EAE INSTRUCTIONS AND A NEW MULTIPLY ALGORITHM TO PROVIDE A SUBSTANTIAL IMPROVEMENT OVER THE REGULAR PATCH SHOWN ON PAGE 89 OF THE LISTING. THE TIME REQUIRED TO MULTIPLY TWO 36 BIT NUMBERS IS NOW ONLY 155 MICROSECONDS COMPARED TO TYPICALLY 700 MICROSECONDS WITH THE OLDER EAE ROUTINE AND OVER 2800 MICROSECONDS WITH SOFTWARE. THE RESULT IS ALSO ROUNDED OFF - NOT JUST TRUNCATED - WHICH IMPROVES THE ACCURACY. THIS IS DEMONSTRATED BY THE COMMAND "T %, 1E99" WHICH NOW GIVES "~~0.1000000000~~7E+100" INSTEAD OF "~~0.999999995~~4E+99".

THE THREE-WORD NORMALIZE ROUTINE HAS ALSO BEEN REWRITTEN. THE ENTIRE NORMALIZATION NOW TAKES ONLY 60 MICROSECONDS. PREVIOUSLY IT TOOK THAT LONG JUST TO DEAL WITH NEGATIVE NUMBERS SINCE THE SOFTWARE (UNLIKE THE HARDWARE) COULD ONLY HANDLE POSITIVE NUMBERS. ADDITIONALLY A SHORTCUT USED BY THE SOFTWARE HAS BEEN ELIMINATED WHICH IMPROVES THE ACCURACY OF CERTAIN NEGATIVE NUMBERS BY A FACTOR OF TWO. ANOTHER ADVANTAGE OF THE HARDWARE ROUTINE IS THAT MULTIPLE SHIFTS ADD ONLY A FRACTION OF A MICROSECOND PER SHIFT. THIS MEANS THAT EVEN THE LONGEST NORMALIZATION (34 SHIFTS) TAKES ONLY 67 MICROSECONDS IN CONTRAST TO THE SOFTWARE WHICH IS 23.5 TIMES SLOWER (1.573 MILLISECONDS). WHILE SUCH DRAMATIC IMPROVEMENTS ARE NOT AS COMMON AS FACTORS OF 2-4, SINCE THE NORMALIZE ROUTINE IS USED AT THE END OF EVERY ARITHMETIC OPERATION THESE IMPROVEMENTS SPEED UP ALL FLOATING POINT CALCULATIONS AND NOT JUST THOSE INVOLVING MULTIPLICATION.

BECAUSE THE HARDWARE DIVIDE INSTRUCTION USES ONLY A SINGLE-WORD DIVISOR (AND BECAUSE DIVISION IS NOT ASSOCIATIVE) THIS INSTRUCTION CANNOT BE USED IN THE FLOATING DIVIDE ROUTINE, BUT IT IS USED IN TWO OTHER PLACES TO CREATE 8 WORD "HOLES". FOR FLOATING DIVISION TWO DISTINCT APPROACHES HAVE BEEN USED: ONE FOR THE STANDARD VERSION OF FOCAL AND THE OTHER FOR THE 10-DIGIT VERSION. THE LATTER OFFERS ONLY MINOR IMPROVEMENTS UPON THE ROUTINE SHOWN IN THE FOCAL LISTING, BUT THE REGULAR VERSION NOW TAKES ADVANTAGE OF THE DAD INSTRUCTION TO REDUCE THE DIVIDE TIME BY A FACTOR OF 1.7 (TYPICALLY). IN BOTH CASES THE "ZERO DIVISOR" ERROR CODE HAS BEEN CHANGED TO 28.<1 AS A RESULT OF MOVING THINGS AROUND TO CONSOLIDATE EMPTY AREAS.

THERE ARE SEVERAL OTHER PLACES WHERE EAE INSTRUCTIONS (OR EVEN JUST THE BSW INSTRUCTION) COULD BE USED. HOWEVER, SINCE GENERALLY ONLY A FEW MICROSECONDS WOULD BE SAVED, THESE CHANGES WERE FELT TO BE INSIGNIFICANT. ONE EXCEPTION IS THE ALIGN/FIX ROUTINE WHICH COULD GREATLY BENEFIT FROM THE ASR INSTRUCTION. UNFORTUNATELY THE MODIFICATION WOULD REQUIRE ADDITIONAL CORE SPACE IN A TIGHTLY-CODED AREA AND SO WAS NOT INCLUDED.

LASTLY IT SHOULD BE NOTED THAT BEFORE EXITING FROM THE FLOATING-POINT INTERPRETER THE MQ REGISTER IS CLEARED AND THE EAE RETURNED TO MODE A. THUS OTHER ROUTINES, ESPECIALLY THOSE WRITTEN FOR PREVIOUS EAE'S, MAY BE USED WITHOUT CONCERN AS TO WHICH MODE IS SET.

THE BINARY TAPE CONTAINS TWO SECTIONS: THE FIRST LOADS THE 10-DIGIT VERSION OF ALL ROUTINES AND THE SECOND MODIFIES THE MULTIPLY AND DIVIDE ROUTINES FOR THE NORMAL (3 WORD) VERSION. THUS FOR STANDARD FOCAL THE ENTIRE TAPE SHOULD BE READ IN AFTER LOADING FOCAL. WHEN THE 4-WORD (10 DIGIT) OVERLAY IS USED, IT SHOULD BE LOADED FIRST, FOLLOWED BY THE FIRST SECTION (ONLY) OF THE EAE TAPE. THERE IS NO KNOWN CONFLICT WITH OTHER FOCAL OVERLAYS.

THE TOTAL FREE SPACE AVAILABLE DEPENDS ON WHETHER THE 6- OR 10-DIGIT VERSION IS USED. WITH THE 10-DIGIT VERSION THE LARGEST FREE BLOCK IS 46 WORDS LONG. IF ONLY THE 6-DIGIT VERSION IS REQUIRED, THIS PARTICULAR AREA BECOMES 49 WORDS LONG AND ANOTHER 45 WORD BLOCK IS ALSO AVAILABLE. (THIS COULD BE 48 WORDS IF "MINUS" WERE SIMPLIFIED.) THE LOCATION OF STILL OTHER FREE AREAS CAN BE DETERMINED FROM THE LISTING. SINCE THE REGULAR EAE PATCH SAVED ONLY 12 WORDS THIS FEATURE MAY BE CONSIDERED A BIG IMPROVEMENT TOO. THE AUTHOR, FOR INSTANCE, HAS IMPLEMENTED HIS FCOM FUNCTION IN PART OF THIS SPACE.

THE FOLLOWING COMPARATIVE EXECUTION TIMES (IN MICROSECONDS) MAY BE OF INTEREST. THE MEASUREMENTS DO NOT INCLUDE THE TIME REQUIRED FOR SIGN-CHECKING OR NORMALIZATION, NOR DO THEY INCLUDE THE ARGUMENT FETCH - ALL OF WHICH NATURALLY REDUCE THE NET IMPROVEMENT IN PROGRAM SPEED. AS AN OVERALL FIGURE, IT WAS OBSERVED THAT A PROGRAM PERFORMING A LEAST-SQUARES FIT ON 1022 DATA POINTS (DATA WAS STORED ON DECTAPE IN ASCII FORMAT) RAN 14% SLOWER WITH THE REGULAR EAE PATCH AND 25% SLOWER WITH JUST THE SOFTWARE ROUTINES (10 DIGIT VERSIONS). THIS RESULT IS PROBABLY TYPICAL FOR CALCULATION-BOUND PROGRAMS.

3 WORD VERSION

4 WORD VERSION

N*N: N ⁿ		1	2 ¹² -1	2 ²³ -1	PI	1	2 ¹² -1	2 ²³ -1	2 ²³ +3	2 ³⁵ -1	PI
A)	SOFTWARE	660	1281	1345	1292	1309	2550	2670	2730	3004	2864
B)	STD. EAE	207	321	327	327	401	629	636	686	713	699
C)	THIS EAE	68	68	68	68	155	155	155	155	158	155
RATIO A/B		3.2	4.0	4.1	4.0	3.3	4.0	4.2	4.0	4.2	4.1
RATIO A/C		9.7	18.8	19.8	19.0	8.4	16.5	17.2	17.6	19.0	18.5

/FOCAL PATCHES FOR 8/E EAE JVZ

/THESE PATCHES SUBSTANTIALLY REWRITE FOCAL'S MULTIPLY AND /NORMALIZE ROUTINES IN ORDER TO USE EAE INSTRUCTIONS. THE /HARDWARE ASSUMED IS A PDP-8/E EXTENDED ARITHMETIC ELEMENT /TYPE KE8-E. MODE B INSTRUCTIONS ARE USED EXTENSIVELY. THE /MEASURED EXECUTION TIME FOR A 36 BIT BY 36 BIT MULTIPLY /IS ONLY 155 MICROSECONDS - ABOUT HALF THE TIME NEEDED FOR /ONE 12 BIT PRODUCT (OF WHICH THERE ARE 9) USING THE STAN- /DARD SOFTWARE ROUTINES AND FASTER BY A FACTOR OF 4.5 THAN /THE PATCH SHOWN ON P. 89 OF THE FOCAL LISTING. SIMILARLY /THE TIME FOR A 1 BIT NORMALIZE HAS BEEN DECREASED BY A /FACTOR OF 2. FOR NEGATIVE NUMBERS A FACTOR OF 3 IS OB- /TAINED SINCE NO INVERSION IS REQUIRED. DIVIDE AND SOME /OTHER ROUTINES HAVE ALSO BEEN REWRITTEN IN ORDER TO CON- /SOLIDATE EMPTY AREAS. THESE MODIFICATIONS MAKE AVAILABLE /78 WORDS AND CHANGE THE ZERO-DIVISOR ERROR CODE TO 28.<1.

/SYSTEM SYMBOLS:

7405	MUY=7405
7407	DVI=7407
7411	NMI=7411
7413	SHL=7413
7421	MQL=7421
7431	SWAB=7431
7441	SCA=7441
7445	DST=7445
7447	SWBA=7447
7451	DPSZ=7451
7521	SWP=7521
7575	DCM=7575
7621	CAM=7621
7701	ACL=7701

FIXTAB

/FOCAL SYMBOLS:

0033	T3=33	
0040	FLOP=40	/OPERAND
0044	FLAC=44	/ACCUMULATOR
0050	SIGNF=50	/SIGN OF RESULT
0113	C260=113	
0124	P4000=124	
2442	PRNT=2442	
6000	FLOUTP=6000	
6136	SMIN=6136	
6154	OUTDG=6154	
6400	FPNT=6400	
6520	ZERO=6520	

/FOCAL INSTRUCTIONS:

	4451	NEGATE=4451	/JMS I 51
	4527	SHIFTL=4527	/JMS I 127
	4551	PRINTC=4551	/JMS I 151
	4566	ERROR2=4566	/JMS I 166
	2444	*2444	/PRINT TWO DECIMAL DIGITS
02444	7427	SQL DVI	/DIVIDE BY TEN
02445	0012	12	
02446	7521	SWP	/GET QUOTIENT
02447	1113	TAD C260	/FORM ASCII
02450	4551	PRINTC	
02451	7521	SWP	/GET REMAINDER
02452	1113	TAD C260	
02453	4551	PRINTC	
02454	5642	JMP I PRNT	
02455	0000	ZBLOCK 10	/EIGHT LEFT OVER
	6104	*6104	/EXPONENT OUTPUT ROUTINE
06104	7700	SMA CLA	/TEST SIGN OF EXPONENT
06105	7144	CMA CLL RAL	/CREATE "+"
06106	1336	TAD SMIN	/OR "-"
06107	4551	PRINTC	
06110	1033	TAD T3	/GET DECIMAL EXPONENT
06111	7510	SPA	
06112	7041	CIA	/TAKE ABSOLUTE VALUE
06113	7427	SQL DVI	/DIVIDE BY ONE HUNDRED
06114	0144	144	
06115	7521	SWP	/PRINT QUOTIENT
06116	7440	SZA	/UNLESS IT'S ZERO
06117	4354	JMS OUTDG	
06120	7521	SWP	/NOW PRINT REMAINDER
06121	4737	JMS I PRNTI	
06122	5600	JMP I FLOUTP	
06123	0000	ZBLOCK 10	/EIGHT FREE

/POINTERS REQUIRING CHANGES:

06137	6137	*6137	
06137	2442	PRNTI, PRNT	
06304	6304	*6304	
06304	7160	RESOL5, RESOLV	/FOR FLOATING INPUT
06565	6565	*6565	
06565	7123	OPMINS, MINUS	/FOR SUBTRACTION
06576	6576	*6576	
06576	7165	FLDV	/FOR DIVISION
06752	6752	*6752	
06752	7160	RESOL, RESOLV	/FOR FIX

/THREE WORD BY THREE WORD UNSIGNED MULTIPLY ROUTINE

/ (A+B+C)*(D+E+F) = (ONE+TWO+THREE+FOUR+FIVE+SIX)

/THE METHOD GENERATES ALL 6 WORDS, BUT IS MODIFIED
 /HERE TO PRESERVE ONLY THE 3 MOST SIGNIFICANT ONES.
 /ADDITIONALLY THE RESULT IS ROUNDED OFF RATHER THAN
 /SIMPLY TRUNCATED WHICH IMPROVES THE ACCURACY.

07003	7003		*7003	
07003	7735		-43	/LOOP COUNT TERMINATES LIST
07004	0000	MULT,	0	
07005	1040		TAD EX1	
07006	4340		JMS SIGN	/ADD EXPONENTS (PLUS 1)
07007	7450		SNA	/TEST FOR ZERO OPERAND
07010	5757		JMP I NULL	
07011	7710		SPA CLA	/POSITIVE OPERAND REQUIRED
07012	4323		JMS MINUS	
07013	7431		SWAB	/SET MODE B
07014	1047	CF,	TAD C	/(TAD P4000) FOR 2X2
07015	7425		MQL MUY	/(JMP BE-1) FOR 2X2
07016	0043		F	
07017	7421		MQL	/SAVE HIGH ORDER & ERASE SIX
07020	1046	BF,	TAD B	
07021	7525		SWP MUY	/USE PREVIOUS HIGH ORDER AS
07022	0043		F	/REMAINDER IN THIS POSITION
07023	1124		TAD P4000	/ROUND OFF
07024	3360		DCA FOUR	/SAVE INTERMEDIATE RESULT
07025	7004		RAL	
07026	3340		DCA THREE	/SAVE CARRY AND/OR CLEAR THREE
07027	1047	CE,	TAD C	
07030	7525		SWP MUY	/ADD IN PREVIOUS
07031	0042		E	/PARTIAL PRODUCT
07032	1360		TAD FOUR	/SUM HIGH ORDER PARTS
07033	7421		MQL	/DISCARD FIVE AND SAVE CARRY
07034	7430		SZL	
07035	2340		ISZ THREE	/ACCUMULATE CARRIES
07036	1045	AF,	TAD A	
07037	7525		SWP MUY	
07040	0043		F	
07041	1340		TAD THREE	/BUILD UP LEAST SIGNIFICANT
07042	3340		DCA THREE	/PART OF FINAL RESULT
07043	1047	CD,	TAD C	
07044	7525		SWP MUY	
07045	0041		D	
07046	1340		TAD THREE	
07047	3340		DCA THREE	/ENTRY POINT FOR 2X2

/FOCAL PATCHES FOR 8/E EAE JVZ PAL8-V7 6/7/73

07050	1046	BE,	TAD B	
07051	7525		SWP MUY	
07052	0042		E	
07053	1340		TAD THREE	
07054	7421		MQL	/DISCARD FOUR
07055	7004		RAL	
07056	3323		DCA TWO	/SAVE CARRY
07057	1045	AE,	TAD A	
07060	7525		SWP MUY	
07061	0042		E	
07062	1323		TAD TWO	/SUM CARRY BITS
07063	3323		DCA TWO	/WITH CARRY WORD
07064	1046	BD,	TAD B	
07065	7525		SWP MUY	
07066	0041		D	
07067	1323		TAD TWO	
07070	7521		SWP	/(MQL) FOR 2X2
07071	3047		DCA OVER2	/SAVE THREE
07072	1045	AD,	TAD A	
07073	7525		SWP MUY	
07074	0041		D	
07075	7565		SWP DST	/SAVE ONE & TWO
07076	0045		HORD	
07077	4702		JMS I NORM	/NORMALIZE IF REQUIRED
07100	4360		JMS RESOLV	/RESTORE PROPER SIGN
07101	5604		JMP I MULT	/AND RETURN
07102	7335	NORM,	DNORM	
07103	0000		ZBLOCK 20	/SIXTEEN FREE WORDS

/STANDARD DEFINITIONS:

0040	EX1=FLOP
0041	AC1H=FLOP+1
0042	AC1L=FLOP+2
0043	OVER1=FLOP+3

0044	EXP=FLAC
0045	HORD=FLAC+1
0046	LORD=FLAC+2
0047	OVER2=FLAC+3

/LETTER ASSIGNMENTS:

0045	A=HORD	/FLAC
0046	B=LORD	
0047	C=OVER2	

0041	D=AC1H	/OPERAND
0042	E=AC1L	
0043	F=OVER1	

/THESE SUBROUTINES PREPARE MULTIPLY AND DIVIDE FOR
 /ANY COMBINATION OF SIGNED ARGUMENTS, OR FOR ZERO.
 /THE RESULT OF EITHER IS ZERO IF FLAC IS ZERO; IF
 /THE OPERAND IS ZERO THE PRODUCT IS ALSO ZERO, BUT
 /ATTEMPTED DIVISION BY ZERO RETURNS WITH AN ERROR.

/THE SIGN ROUTINE IS ENTERED WITH THE OPERAND EXPO-
 /NENT IN THE AC AND EXITS WITH THE SIGN OF THE OP-
 /ERAND THERE TO FACILITATE FURTHER TESTING. THE
 /SIGN OF THE RESULT IS STORED IN SIGNF.

07123	0000	MINUS,	0	/NEGATES OPERAND
07124	1043		TAD OVERI	
07125	7431		SWAB	
07126	1042		TAD ACIL	
07127	7575		DCM	
07130	7565		SWP DST	
07131	0042		ACIL	
07132	7447		SWBA	
07133	7224		CLA CML RAL	
07134	1041		TAD ACIH	
07135	7041		CIA	
07136	3041		DCA ACIH	
07137	5723		JMP I MINUS	
07140	0000	SIGN,	0	
07141	7001		IAC	/ADD 1 TO EXPONENT OF OP.
07142	1044		TAD EXP	/COMPUTE EXPONENT OF RESULT
07143	3044		DCA EXP	
07144	7130		STL RAR	/SET BIT 0
07145	0041		AND ACIH	/EXTRACT SIGN
07146	1045		TAD HORD	/FORM XOR OF SIGNS
07147	3050		DCA SIGNF	/SIGN OF RESULT
07150	1045		TAD HORD	
07151	7450		SNA	/TEST IF RESULT = ZERO
07152	5757		JMP I NULL	
07153	7710		SPA CLA	/ABSOLUTE VALUE REQUIRED
07154	4451		NEGATE	
07155	1041		TAD ACIH	/CHECK THE OPERAND
07156	5740		JMP I SIGN	/FOR VARIOUS THINGS
07157	6520	NULL,	ZERO	
07160	0000	RESOLV,	0	/SET PROPER SIGN
07161	1050		TAD SIGNF	
07162	7710		SPA CLA	
07163	4451		NEGATE	
07164	5760		JMP I RESOLV	

/NUMBER ASSIGNMENTS:

7123	TWO=MINUS
7140	THREE=SIGN
7160	FOUR=RESOLV

/THIS IS THE FLOATING DIVIDE ROUTINE (THREE WORDS)

07165	1040	FLDV,	TAD EX1	/GET OPERAND EXPONENT
07166	7041		CIA	/AND NEGATE
07167	4340		JMS SIGN	/CHECK SIGNS & ZERO
07170	7450		SNA	/CHECK FOR ZERO DIVISOR
07171	4566		ERROR2	/THAT'S A NO-NO!
07172	7700		SMA CLA	
07173	4323		JMS MINUS	/NEGATIVE OPERAND REQUIRED
07174	1203		TAD MULT-1	/SET UP LOOP COUNT
07175	3040		DCA EX1	
07176	7431		SWAB	/CLEAR RESULT
07177	7410		SKP	
07200	4527	DLOOP,	SHIFTL	/SHIFT FLAC LEFT
07201	7100		CLL	/THREE WORD SUBTRACTION
07202	1043		TAD OVER1	
07203	1047		TAD OVER2	
07204	3335		DCA DNORM	/TEMPORARY STORAGE
07205	7004		RAL	
07206	1042		TAD AC1L	
07207	1046		TAD LORD	
07210	3364		DCA SCNT	
07211	7004		RAL	
07212	1041		TAD AC1H	
07213	1045		TAD HORD	
07214	7420		SNL	/LINK=1 IF SUCCESSFUL (EG >0)
07215	5223		JMP .+6	
07216	3045		DCA HORD	/SAVE RESULT
07217	1364		TAD SCNT	
07220	3046		DCA LORD	
07221	1335		TAD DNORM	
07222	3047		DCA OVER2	
07223	7701		ACL	/SAVE LINK BITS IN QHI, QLO, MQ
07224	7004		RAL	
07225	7421		SQL	
07226	1377		TAD QLO	
07227	7004		RAL	
07230	3377		DCA QLO	
07231	1376		TAD QHI	
07232	7004		RAL	
07233	3376		DCA QHI	
07234	2040		ISZ EX1	/INCREMENT STEP COUNTER
07235	5200		JMP DLOOP	
07236	1376		TAD QHI	/SAVE RESULT AT THE END
07237	3045		DCA HORD	
07240	1377		TAD QLO	
07241	7565		SWP DST	
07242	0046		LORD	
07243	4335		JMS DNORM	/NORMALIZE IF NECESSARY
07244	4647		JMS I RSLV	/SET PROPER SIGN
07245	5646		JMP I .+1	
07246	6401		FPNT+1	
07247	7160	RSLV,	RESOLV	

/FOCAL PATCHES FOR 8/E EAE JVZ PAL8-V7 6/7/73

```

07250 0000          ZBLOCK 56          /FREE SPACE

          /EAE SIGNED NORMALIZE ROUTINE: (THREE WORDS)

07326 7411  GT13,  NMI          /NORMALIZE LOWER PART
07327 7565          SWP DST
07330 0045          HORD
07331 7661          CAM SCA          /GET NEW SHIFT COUNT
07332 1334          TAD P14          /PLUS TWELVE
07333 5367          JMP ALL0-2
07334 0014  P14,    14

07335 0000  DNORM,  0
07336 7300          CLA CLL          /CLEAR LINK FOR LATER
07337 1046          TAD LORD
07340 7421          MQL          /LOAD AC, MQ
07341 1045          TAD HORD
07342 7447          SWBA          /SET MODE A TO
07343 7411          NMI          /PROTECT 4000 0000
07344 7431          SWAB          /SAVE RESULT IN MQ
07345 7441          SCA          /GET SHIFT COUNT
07346 7451          DPSZ          /CHECK FOR ZERO
07347 7141          CLL CIA
07350 1334          TAD P14          /SET LINK IF < TWELVE
07351 7641          CLA SCA
07352 3364          DCA SCNT          /SAVE SHIFT COUNT
07353 1047          TAD OVER2
07354 7571          SWP DPSZ          /CHECK FOR ZERO
07355 7410          SKP          /AC=RESULT
07356 5371          JMP ALL0

07357 3045          DCA HORD
07360 1046          TAD LORD
07361 7420          SNL          /.GE. TWELVE ?
07362 5326          JMP GT13          /YES: IGNORE SIGN BITS
07363 7413          SHL          /MODE B
07364 0000  SCNT,  0
07365 3046          DCA LORD
07366 1364          TAD SCNT          /CORRECT EXPONENT
07367 7041          CIA
07370 1044          TAD EXP
07371 3044  ALL0,  DCA EXP
07372 7521          SWP          /CLEAR MQ
07373 3047          DCA OVER2          /SAVE OR CLEAR LAST WORD
07374 7447          SWBA          /RESTORE MODE A
07375 5735          JMP I DNORM

07376 0000  QHI,  0
07377 0000  QLO,  0

```

/THESE OVERLAYS MODIFY THE EAE MULTIPLY & DIVIDE ROUTINES
 /FOR MAXIMUM SPEED WITH 24 BIT MANTISSAS. THE MULTIPLY
 /TIME IS LESS THAN 70 MICROSECONDS AND THE DIVIDE TIME IS
 /ABOUT 950 MICROSECONDS (A FACTOR OF 1.7 IMPROVEMENT). AN
 /ADDITIONAL 35 WORDS ARE AVAILABLE (TOTAL OF 113).

/ADDITIONAL EAE INSTRUCTIONS:

7443 DAD=7443
 7663 DLD=7663

/TWO WORD BY TWO WORD UNSIGNED MULTIPLY ROUTINE

/(A+B)*(D+E) = (ONE+TWO+THREE+FOUR)

/THE RESULT IS ROUNDED TO THE TWO MOST SIGNIFICANT WORDS.

07003	7003		*7003	
	7751		-27	/NEW LOOP COUNT
07014	7014	EB,	*7014	
07015	1042		TAD E	
07016	7425		MQL MUY	
07017	0046		B	
	7421		MQL	/DISCARD FOUR
07020	1042	EA,	TAD E	
07021	7525		SWP MUY	/USE PREVIOUS HIGH ORDER AS
07022	0045		A	/REMAINDER IN THIS POSITION
07023	3323		DCA TWO	/SAVE INTERMEDIATE RESULT
07024	1041	DB,	TAD D	
07025	7525		SWP MUY	/ADD IN PREVIOUS
07026	0046		B	/PARTIAL PRODUCT
07027	7521		SWP	
07030	7710		SPA CLA	/CHECK AND DISCARD THREE
07031	2323		ISZ TWO	/ROUND OFF
07032	7501		MQA	
07033	1323		TAD TWO	/SUM HIGH ORDER PARTS AND
07034	7421		MQL	/ ADD INTO LAST PRODUCT
07035	1041	DA,	TAD D	
07036	7525		SWP MUY	
07037	0045		A	
07040	7565		SWP DST	/SAVE FINAL RESULT
07041	0045		HORD	
07042	4645		JMS I NORM	/NORMALIZE
07043	4360		JMS RESOLV	/RESTORE PROPER SIGN
07044	5604		JMP I MULT	/AND RETURN
07045	7335	NORM,	DNORM	
07046	0000		ZBLOCK 35	/EXTRA FREE SPACE

/CHANGES TO FLOATING DIVIDE FOR TWO WORDS:

	7176		*7176	
07176	1042		TAD ACIL	/GET DIVISOR
07177	7431		SWAB	
07200	1041		TAD ACIH	
07201	7445		DST	/REVERSE WORD ORDER
07202	0041		ACIH	
07203	7663		DLD	/GET DIVIDEND
07204	0046		LORD	
07205	3046		DCA LORD	/CLEAR QUOTIENT
07206	1045		TAD HORD	
07207	5214		JMP .+5	
07210	7663	DLUP,	DLD	/GET OPERAND
07211	7376		QHI	
07212	7413		SHL	/SHIFT LEFT
07213	0001		I	
07214	7445		DST	/SAVE FOR NEXT TIME
07215	7376		QHI	
07216	7443		DAD	/SUBTRACT FLOP
07217	0041		ACIH	
07220	7420		SNL	/TEST FOR OVERFLOW
07221	5224		JMP .+3	
07222	7445		DST	/SAVE NEW RESULT
07223	7376		QHI	
07224	7621		CAM	
07225	1046		TAD LORD	/SAVE LINK BITS
07226	7004		RAL	
07227	3046		DCA LORD	
07230	1045		TAD HORD	
07231	7004		RAL	
07232	3045		DCA HORD	
07233	2040		ISZ EX1	/TEST LOOP COUNT
07234	5210		JMP DLUP	
07235	4335		JMS DNORM	/NORMALIZE
07236	4640		JMS I .+2	/SET PROPER SIGN
07237	5641		JMP I .+2	/RETURN
07240	7160		RESOLV	
07241	6401		FPNT+1	
07242	0000		ZBLOCK 6	/EXTRA
			S	

A	0045	QHI	7376
ACIH	0041	QLO	7377
ACIL	0042	RESOL	6752
AD	7072	RESOLV	7160
AE	7057	RESOL5	6304
AF	7036	RSLV	7247
ALL0	7371	SCNT	7364
B	0046	SHIFTL	4527
BD	7064	SIGN	7140
BE	7050	SIGNF	0050
BF	7020	SMIN	6136
C	0047	THREE	7140
CD	7043	TWO	7123
CE	7027	T3	0033
CF	7014	ZERO	6520
C260	0113		
D	0041		
DA	7035		
DAD	7443		
DB	7024		
DL D	7663		
DLOOP	7200		
DLUP	7210		
DNORM	7335		
E	0042		
EA	7020		
EB	7014		
ERROR2	4566		
EXP	0044		
EX1	0040		
F	0043		
FLAC	0044		
FLDV	7165		
FLOP	0040		
FLOUTP	6000		
FOUR	7160		
FPNT	6400		
GT13	7326		
HORD	0045		
LORD	0046		
MINUS	7123		
MULT	7004		
NEGATE	4451		
NORM	7102		
NULL	7157		
N0RM	7045		
OPMINS	6565		
OUTDG	6154		
OVER1	0043		
OVER2	0047		
PRINTC	4551		
PRNT	2442		
PRNT1	6137		
P14	7334		
P4000	0124		